

Maticové vzorce III



Nebaví vás neustále měnit oblasti, které má Excel sčítat, zkuste maticové vzorce ve spojení s funkcemi NEPŘÍMÝ.ODKAZ a ODKAZ.

V druhém dílu našeho miniseriálu [Maticové vzorce II](#), jsme prozkoumali možnosti maticových vzorců na poli podmíněných součtů a počtů. Dnes si tuto oblast mírně rozšíříme a podíváme se na to, jak dynamicky měnit sčítanou oblast; samozřejmě zachováme podmíněný součet.

O co vlastně jde...

Používáte-li funkci SUMIFS, musíte vždy definovat, kterou oblast budete sčítat. Maticové vzorce dokáží měnit sčítanou oblast (nebo i oblasti podmínek) na základě hodnoty v buňce, což může být mnohdy velice praktické (např. při aplikování maticových vzorců na výjezdy z databází, které nemají požadované sloupce vždy na stejném místě, tedy někdy ve **sloupci C** a podruhé např. ve **sloupci F**).

Podívejme se na obrázek níže, který nápadně připomíná naši úlohu z minulého dílu, jen jsme přidali údaje o počtu zákazníků a časové náročnosti prodeje. Jak už zřejmě tušíte, budeme schopni napsat takový vzorec, který splní, jak kritéria v buňkách **H2:H5**, tak v závislosti na hodnotě v buňce **J1** sečte:

- Částku.
- Počet zákazníků.
- Strávený čas.

a to vše bez zásahů do vzorce, pouze na základě změny vstupních údajů v buňkách.

J2		fx {=SUMA((A2:A20>=H3)*(A2:A20<=H4)*(B2:B20=H2)*(C2:C20=H5)*(NEPŘÍMÝ.ODKAZ(ODKAZ(2;POZVYHLEDAT(J1;A1:G1;0)) & ":" & ODKAZ(20;POZVYHLEDAT(J1;A1:G1;0))))}}								
	A	B	C	D	E	F	G	H	I	J
1	Datum	Obchodník	Oblast	Částka	Počet zákazníků	Strávený čas (hod)		Kritéria		Strávený čas (hod)
2	3.1.2012	Jirka	sport	2 500	5	2	jméno	Jirka		5
3	15.1.2012	Jirka	kultura	1 200	10	1	od	1.1.2011		
4	30.1.2012	Jirka	kultura	300	3	3	do	31.3.2012		
5	5.2.2012	Jirka	sport	230	1	1	oblast	sport		
6	7.2.2012	David	sport	450	2	0,5				
7	28.2.2012	David	sport	1 450	1	0,25				
8	3.3.2012	Jirka	sport	2 510	10	2				
9	10.3.2012	David	kultura	320	2	0,25				
10	21.3.2012	David	kultura	480	1	0,5				
11	31.3.2012	David	sport	420	1	0,1				
12	8.4.2012	Jirka	kultura	360	6	1				
13	12.4.2012	David	kultura	5 600	4	2				
14	25.5.2012	Jirka	sport	1 000	20	4				

V řádku vzorců vidíte zmíněný vzorec - vypadá strašlivě, ale rozebereme si ho postupně a uvidíte, že je v něm jen jediná novinka oproti minulému dílu.

Vezmeme to ale hezky od začátku - z přechozích dílů víte, proč vzorec začíná a končí složenou závorkou a že se tyto závorky nepíší (Excel je doplní sám po ukončení vzorce stiskem **CTRL + SHIFT + ENTER**, protože jde o maticový vzorec).

1. Úvodní SUMA naznačuje, že celý tento vzorec budeme sčítat.
2. Výraz: (A2:A20>=H3)*(A2:A20<=H4) říká, že chceme sčítat jen prodeje, které se uskutečnily v období od 1.1.2011 do 31.3.2012 včetně.
3. Výraz: (B2:B20=H2) zajistí, že budeme sčítat pouze prodeje obchodníka Jirky.
4. Výraz: (C2:C20=H5) je zde proto, abychom sečetli pouze prodeje v oblasti sport.

A nyní se dostáváme k samotné novince tohoto dílu

```
(NEPŘÍMÝ.ODKAZ(ODKAZ(2;POZVYHLEDAT(J1;A1:G1;0))
```

```
&" ":"&
```

```
ODKAZ(20;POZVYHLEDAT(J1;A1:G1;0)))
```

Abychom tomuto porozuměli, uděláme slavný Cimrmanovský krok stranou a podíváme se na funkce NEPŘÍMÝ.ODKAZ a ODKAZ.

Nepřímý odkaz

Tato funkce je velice platná, protože dokáže z textového řetězce (z textu) udělat odkaz. Zřejmé to bude na následujícím obrázku.

C5		fx		=NEPŘÍMÝ.ODKAZ(A1)	
	A	B	C	D	E
1	B3	(v buňce A1 je prostým textem napsáno B3)			
2					
3		obsah buňky B3			
4					
5			obsah buňky B3	=NEPŘÍMÝ.ODKAZ(A1)	
6			obsah buňky B3	=NEPŘÍMÝ.ODKAZ("B3")	

V buňce **A1** je text „B3“, v buňce **B3** je text „obsah buňky B3“ a buňce **C5** je vzorec =NEPŘÍMÝ.ODKAZ(A1), se stejným výsledkem by skončil vzorec =NEPŘÍMÝ.ODKAZ("B3"), protože obsah buňky **A1** je „B3“.

Odkaz

Tato funkce je vlastně opakem funkce NEPŘÍMÝ.ODKAZ, protože vrací adresu buňky podle zadaných hodnot řádku a sloupce. Funkce ODKAZ, tedy přijímá dva argumenty =ODKAZ(řádek, sloupec) a tyto číselné hodnoty do ní lze vložit jak přímým zápisem čísla, tak odkazem na buňku. Jak je vidět v následujícím obrázku.

B4		fx		=ODKAZ(B1;B2)	
	A	B	C	D	
1	řádek	10			
2	sloupec	5			
3					
4		ŠEŠ10	=ODKAZ(B1;B2)		
5		ŠEŠ10	=ODKAZ(10;5)		

V buňce **B4** se funkce odkazuje na buňky **B1** (řádek) a **B2** (sloupec), kdežto v buňce **B5** jsou hodnoty zapsány přímo do funkce. V obou případech je však výsledek stejný, tedy textový odkaz na buňku **E10**.

Zpět k našemu maticovému vzorci

Nyní už víme, že funkce NEPŘÍMÝ.ODKAZ přijímá text a udělá z něj odkaz. Podívejme se tedy znovu na celý závěrečný výraz vzorce:

```
(NEPŘÍMÝ.ODKAZ(ODKAZ(2;POZVYHLEDAT(J1;A1:G1;0))
```

```
&" ":"&
```

```
ODKAZ(20;POZVYHLEDAT(J1;A1:G1;0)))
```

Zvýrazněné části výrazu představují text, ale jaký?

- Podíváme-li se na oba tvary funkce ODKAZ, zjistíme, že jediný rozdíl je v první části (určení řádku) – tedy první funkci ODKAZ je předán parametr **2** a v druhém výskytu parametr **20**
2 a 20 jsme zadali proto, že i ostatní matice máme od řádku 2 do řádku 20 (funkce POZVYHLEDAT je stejná u obou výskytů funkce ODKAZ)
- Funkce POZVYHLEDAT potom vrátí pořadí v oblasti **A1:G1**, kde byla nalezena hodnota zapsaná v buňce **J1** („Strávený čas (hod)“) -> ta je ve **sloupci F**, kterému odpovídá číslo **6** (A=1, B=2...F=6)

3. Mezi oběma **ODKAZ**Y je ještě tento zápis: **& ":" &** - ampersand (&) se používá pro spojování textů a přímo zapsaný text musí být v uvozovkách.
4. Výsledkem bude odkaz na oblast **F2:F20** - takže změnou buňky **J1** měníme i oblast sčítání.
5. Pokud bychom totiž do buňky **J1** napsali slovo „Částka“ byla by výsledná oblast **D2:D20**

Na závěr bych jenom rád připomněl, že maticové vzorce je možné kombinovat s velkou spoustou tradičních funkcí, jako jsou **SMALL**, **LARGE**, **ŘÁDEK**, **SLOUPEC** a s mnohými dalšími. Na to se podíváme v dalším dílu našeho miniseriálu.

Soubor s příklady maticových vzorců z článku [Nejste přihlášen\(a\)](#).

Autor: [Poki](#) • Vydáno: 24.9.2013 13:50 • Přečteno: 14813x